Kubische Kostenfunktion

Dokumentnummer: DX1583
Fachgebiet: Analysis, Kosten- und Preistheorie
Einsatz: 3HAK (zweites Lernjahr)

1 Aufgabe

Berechne
a) das Betriebsoptimum,
b) die langfristige Preisuntergrenze,
c) die Kostenkehre,
d) die Gewinnzone,
e) den maximalen Gewinn.

Figure 1: Die Ausgangslage, mit Geogebra dargestellt.



2 Lösung

2.1 Bestimmung einer geeigneten Kostenfunktion

```
--> x1:0;K1:2000;
    x2:400;K2:4000;
    x3:900;K3:5000;
    x4:1300;K4:7000;
```
(%o92) 0
(%o93) 2000
(%o94) 400
(%o95) 4000
(%o96) 900
(%o97) 5000
(%o98) 1300
(%o99) 7000

```
--> g(x,K):=K=a*x**3+b*x**2+c*x+d;
```
(%o100) $g(x,K):=K=a\,x^3+b\,x^2+c\,x+d$

```
--> g1:g(x1,K1);
    g2:g(x2,K2);
    g3:g(x3,K3);
    g4:g(x4,K4);
```
(%o101) $2000=d$
(%o102) $4000=d+400\,c+160000\,b+64000000\,a$
(%o103) $5000=d+900\,c+810000\,b+729000000\,a$
(%o104) $7000=d+1300\,c+1690000\,b+2197000000\,a$

```
--> l:algsys([g1,g2,g3,g4],[a,b,c,d]);
```
(%o105) $[[a=\frac{1}{195000},b=-\frac{1}{100},c=\frac{319}{39},d=2000]]$

```
--> Kostenfunktion:g(x,K),l;
```
(%o106) $\frac{x^3}{195000}-\frac{x^2}{100}+\frac{319\,x}{39}+2000=\frac{x^3}{195000}-\frac{x^2}{100}+\frac{319\,x}{39}+2000$

Das ist die gesuchte Kostenfunktion

```
--> K:rhs(Kostenfunktion);
```
(%o107) $\frac{x^3}{195000}-\frac{x^2}{100}+\frac{319\,x}{39}+2000$

```
--> K(x):=''K;
```
(%o108) $K(x):=\frac{x^3}{195000}-\frac{x^2}{100}+\frac{319\,x}{39}+2000$

Das ist Kostenfunktion in Maxima-Schreibweise

2.2 Bestimmung von Betriebsoptimum und langfristiger Preisuntergrenze

```
--> D(x):=K(x)/x;
```
(%o109) $D(x):=\frac{K(x)}{x}$

```
--> ab:diff(D(x),x);
```

(%o110) $$\frac{\frac{x^2}{65000}-\frac{x}{50}+\frac{319}{39}}{x}-\frac{\frac{x^3}{195000}-\frac{x^2}{100}+\frac{319\,x}{39}+2000}{x^2}$$

```
--> l:realroots(ab),numer;
```

(%o111) $[x=1128.201014012098]$

```
--> BO:ev(x,l)$
    BO:floor(BO*10+0.5)/10.0;
```

(%o113) 1128.2

Das ist das Betriebsoptimum.

```
--> LPU:D(BO)$
    LPU:floor(LPU*100+0.5)/100.0;
```

(%o115) 5.2

Das ist die langfristige Preisuntergrenze =
Minimum der Durchschnittskosten = der kostendeckende
Preis

```
--> ab2:diff(K(x),x,2);
```

(%o116) $$\frac{x}{32500}-\frac{1}{50}$$

```
--> l:realroots(ab2),numer;
```

(%o117) $[x=650]$

```
--> K(650);
```

(%o118) 4500

Es ist schlecht, den Wert direkt einzusetzen

```
--> xW:ev(x,l);
```

(%o119) 650

```
--> yW:K(xW);
```

(%o120) 4500

```
--> Kostenkehre:[xW,yW];
```

(%o121) [650,4500]

Das ist die sinnvolle Vorgangsweise

2.3 Bestimmung der Gewinnzone

```
--> p:9;
```

(%o122) 9

```
--> U(x):=x*p;
```

(%o123) $U(x):=x\,p$

```
--> g:U(x)=K(x);
```

(%o124) $9\,x=\frac{x^3}{195000}-\frac{x^2}{100}+\frac{319\,x}{39}+2000$

```
--> l:realroots(g),numer;
```

(%o125) $[x=-438.9199676811695, x=460.8456586897373, x=1928.07430896163]$

```
--> GS:ev(x,l[2])$
    GS:floor(GS*10+0.5)/10.0;
```

(%o127) 460.8

```
--> GG:ev(x,l[3])$
    GG:floor(GG*10+0.5)/10.0;
```

(%o129) 1928.1

```
--> Gewinnzone:[GS,GG];
```

(%o130) [460.8,1928.1]

Das ist die Gewinnzone. Der kleinere Wert ist der
Break-Even-Point = Nutzenschwelle = Gewinnschwelle,
der größere Wert ist ist Nutzengrenze = Gewinngrenze.

2.4 Bestimmung der gewinnmaximalen Menge und des maximalen Gewinns

```
--> G(x):=U(x)-K(x);
```

(%o131) $G(x):=U(x)-K(x)$

```
--> ab:diff(G(x),x);
```

(%o132) $-\frac{x^2}{65000}+\frac{x}{50}+\frac{32}{39}$

```
--> l:realroots(ab),numer;
```

(%o133) $[x=-39.8067362010479, x=1339.806736201048]$

```
--> xC:ev(x,l[2])$
    xC:floor(xC*10+0.5)/10.0;
```

(%o135) 1339.8

Das ist die gewinnmaximale Menge

```
--> Gmax:G(xC)$
    Gmax:floor(Gmax*100+0.5)/100.0;
```

(%o137) 4716.49

Das ist der maximale Gewinn